

VDM++ Tutorial at FM'06

Professor **Peter Gorm Larsen**
Engineering College of Aarhus
Computer Technology & Embedded Systems
(pgl@iha.dk)

Agenda

- Part 1(9:00 – 10:30) The VDM++ Language
 - [Introduction](#)
 - [Access Modifiers and Constructors](#)
 - [Instance Variables](#)
 - [Types](#)
 - [Functions](#)
 - [Expressions, Patterns, Bindings](#)
 - [Operations](#)
 - [Statements](#)
 - [Concurrency](#)
 - Part 2 (11:00 – 12:30) [VDMTools and VDM++ examples](#)

Who gives this tutorial?

- **Peter Gorm Larsen**; MSc, PhD
- 18 years of professional experience
 - ½ year with Technical University of Denmark
 - 13 years with IFAD
 - 3,5 years with Systematic
 - 3/4 year with University College of Aarhus
- Consultant for most large defence contractors on large complex projects (e.g. JSF)
- Relations to industry and academia all over the world
- Has written books and articles about VDM
- See <http://home0.inet.tele.dk/pgl/peter.htm> for details



Vienna Development Method




- Invented at IBM's labs in Vienna in the 70's
- VDM-SL and VDM++
 - ISO Standardisation of VDM-SL
 - VDM++ is an object-oriented extension
- Model-oriented specification:
 - Simple, abstract data types
 - Invariants to restrict membership
 - Functional specification:
 - Referentially transparent functions
 - Operations with side effects on state variables
 - Implicit specification (pre/post)
 - Explicit specification (functional or imperative)

Where has VDM++ been used?

- Modeling critical computer systems e.g. for industries such as
 - Avionics
 - Railways
 - Automotive
 - Nuclear
 - Defense
- I have used this industrially for example at:
 - Boeing, Lockheed-Martin (USA)
 - British Aerospace, Rolls Royce, Adelard (UK)
 - Matra, Dassault, Aerospatiale (France)
 - ...

Industrially Inspired Examples

- Chemical Plant Alarm Management System
- A Robot Controller
- A Road Congestion Warning System




Validation Techniques

- **Inspection**: organized process of examining the model alongside domain experts.
- **Static Analysis**: automatic checks of syntax & type correctness, detect unusual features.
- **Testing**: run the model and check outcomes against expectations.
- **Model Checking**: search the state space to find states that violate the properties we are checking.
- **Proof**: use a logic to reason symbolically about whole classes of states at once.

Validation via Animation

Execution of the model through an interface. The interface can be coded in a programming language of choice so long as a *dynamic link* facility (e.g. CORBA) exists for linking the interface code to the model.




Testing can increase confidence, but is only as good as the test set. Exhaustive techniques could give greater confidence.

Agenda

- Part 1(9:00 – 10:30) The VDM++ Language
 - ✓ [Introduction](#)
 - [Access Modifiers and Constructors](#)
 - [Instance Variables](#)
 - [Types](#)
 - [Functions](#)
 - [Expressions, Patterns, Bindings](#)
 - [Operations](#)
 - [Statements](#)
 - [Concurrency](#)
- Part 2 (11:00 – 12:30) [VDMTools and VDM++ examples](#)

VDM++ Class Outline



Access Modifiers

- VDM++ Class Members may have their access specified as **public**, **private** or **protected**.
- The default for all members is **private**
- Access modifiers may not be narrowed e.g. a subclass can not override a public operation in the superclass with a private operation in the subclass.
- **static** modifiers can be used for definitions which are independent of the object state.

Constructors

- Each class can have a number of constructors
- Syntax identical to operations with a reference to the class name in return type
- The return does not need to be made explicitly
- Can be invoked when a new instance of a class gets created

Agenda

- Part 1(9:00 – 10:30) The VDM++ Language
 - ✓ [Introduction](#)
 - ✓ [Access Modifiers and Constructors](#)
 - [Instance Variables](#)
 - [Types](#)
 - [Functions](#)
 - [Expressions, Patterns, Bindings](#)
 - [Operations](#)
 - [Statements](#)
 - [Concurrency](#)
- Part 2 (11:00 – 12:30) [VDMTools and VDM++ examples](#)

Instance Variables (1)

- Used to model attributes
- Consistency properties modelled as invariants

```
class Person
types
  string = seq of char
instance variables
  name: string := [];
  age: int := 0;
  inv 0 <= age and age <= 99;
end Person
```

Instance Variables (2)

- Used to model associations
- Object reference type simply written as the class name, e.g. *Person*
- Multiplicity using VDM data types

```
class Person
  ...
instance variables
  name: string := [];
  age: int := 0;
  employer: set of Company
  ...
end Person
```

```
class Company
  ...
end Company
```

Agenda

- Part 1(9:00 – 10:30) The VDM++ Language
 - ✓ [Introduction](#)
 - ✓ [Access Modifiers and Constructors](#)
 - ✓ [Instance Variables](#)
 - [Types](#)
 - [Functions](#)
 - [Expressions, Patterns, Bindings](#)
 - [Operations](#)
 - [Statements](#)
 - [Concurrency](#)
- Part 2 (11:00 – 12:30) [VDMTools and VDM++ examples](#)

- Basic types

- Boolean
- Numeric
- Tokens
- Characters
- Quotations

- Compound types

- Set types
- Sequence types
- Map types
- Product types
- Composite types
- Union types
- Optional types
- Function types

Invariants can be added

<code>not b</code>	Negation	<code>bool -> bool</code>
<code>a and b</code>	Conjunction	<code>bool * bool -> bool</code>
<code>a or b</code>	Disjunction	<code>bool * bool -> bool</code>
<code>a => b</code>	Implication	<code>bool * bool -> bool</code>
<code>a <=> b</code>	Biimplication	<code>bool * bool -> bool</code>
<code>a = b</code>	Equality	<code>bool * bool -> bool</code>
<code>a <> b</code>	Inequality	<code>bool * bool -> bool</code>

Quantified expressions can also be considered to be basic operators but we will present them together with the other general expressions

Numeric (1)

<code>-x</code>	Unary minus	<code>real -> real</code>
<code>abs x</code>	Absolute value	<code>real -> real</code>
<code>floor x</code>	Floor	<code>real -> int</code>
<code>x + y</code>	Sum	<code>real * real -> real</code>
<code>x - y</code>	Difference	<code>real * real -> real</code>
<code>x * y</code>	Product	<code>real * real -> real</code>
<code>x / y</code>	Division	<code>real * real -> real</code>
<code>x div y</code>	Integer division	<code>int * int -> int</code>
<code>x rem y</code>	Remainder	<code>int * int -> int</code>
<code>x mod y</code>	Modulus	<code>int * int -> int</code>
<code>x ** y</code>	Power	<code>real * real -> real</code>

Numeric (2)

<code>x < y</code>	Less than	<code>real * real -> bool</code>
<code>x > y</code>	Greater than	<code>real * real -> bool</code>
<code>x <= y</code>	Less or equal	<code>real * real -> bool</code>
<code>x >= y</code>	Greater or equal	<code>real * real -> bool</code>
<code>x = y</code>	Equal	<code>real * real -> bool</code>
<code>x <> y</code>	Not equal	<code>real * real -> bool</code>

Product and Record Types

- Product type definition:

$A_1 * A_2 * \dots * A_n$

Construction of a tuple:

mk_(a1, a2, ..., an)

- Record type definition:

A :: selffirst : A1
 selfsec : A2
 ...
 seln : An

Construction of a record:

mk_A(a1, a2, ..., an)

Example Record Definition

A record type could be defined as:

```
Address ::  
    house    : HouseNumber  
    street   : Street  
    town     : PostalTown
```

With field selectors:

```
mk_Address(15, "The Grove", <London>).street
```

Example Tuple Definition

A tuple type could type could be defined as:

```
nat1 * (seq of char) * PostalTown
```

Then fields can be used using the `.#` operator:

```
mk_(12, "Abstraction Avenue", <Manchester>).#2
```

Overview of Set Operators

<code>e in set s1</code>	Membership (\in)	<code>A * set of A -> bool</code>
<code>e not in set s1</code>	Not membership (\notin)	<code>A * set of A -> bool</code>
<code>s1 union s2</code>	Union (\cup)	<code>set of A * set of A -> set of A</code>
<code>s1 inter s2</code>	Intersection (\cap)	<code>set of A * set of A -> set of A</code>
<code>s1 \ s2</code>	Difference (\setminus)	<code>set of A * set of A -> set of A</code>
<code>s1 subset s2</code>	Subset (\subseteq)	<code>set of A * set of A -> bool</code>
<code>s1 psubset s2</code>	Proper subset (\subset)	<code>set of A * set of A -> bool</code>
<code>s1 = s2</code>	Equality ($=$)	<code>set of A * set of A -> bool</code>
<code>s1 <> s2</code>	Inequality (\neq)	<code>set of A * set of A -> bool</code>
<code>card s1</code>	Cardinality	<code>set of A -> nat</code>
<code>dunion s1</code>	Distr. Union (\cup)	<code>set of set of A -> set of A</code>
<code>dinter s1</code>	Distr. Intersection (\cap)	<code>set of set of A -> set of A</code>
<code>power s1</code>	Finite power set (\mathbb{P})	<code>set of A -> set of set of A</code>

Set Comprehensions

- Using predicates to define sets implicitly
- In VDM++ formulated like:
 - $\{element \mid list\ of\ bindings \ \& \ predicate\}$
- The predicate part is optional
- Quick examples:
 - $\{3 * x \mid x : \mathbf{nat} \ \& \ x < 3\}$ or $\{3 * x \mid x \mathbf{in\ set} \{0, \dots, 2\}\}$
 - $\{x \mid x : \mathbf{nat} \ \& \ x < 5\}$ or $\{x \mid x \mathbf{in\ set} \{0, \dots, 4\}\}$

Sequence Operators

<code>hd l</code>	Head	<code>seq1 of A -> A</code>
<code>tl l</code>	Tail	<code>seq1 of A -> seq of A</code>
<code>len l</code>	Length	<code>seq of A -> nat</code>
<code>elems l</code>	Elements	<code>seq of A -> set of A</code>
<code>inds l</code>	Indexes	<code>seq of A -> set of nat1</code>
<code>l1 ^ l2</code>	Concatenation	<code>seq of A * seq of A -> seq of A</code>
<code>conc l1</code>	Distr. conc.	<code>seq of seq of A -> seq of A</code>
<code>l(i)</code>	Seq. application	<code>seq1 of A * nat1 -> A</code>
<code>l ++ m</code>	Seq. modification	<code>seq1 of A * map nat1 to A -> seq1 of A</code>
<code>l1 = l2</code>	Equality	<code>seq of A * seq of A -> bool</code>
<code>l1 <> l2</code>	Inequality	<code>seq of A * seq of A -> bool</code>

Sequence Comprehensions

- Using predicates to define sequences implicitly
- In VDM++ formulated like:
 - $[element \mid numeric\ set\ binding \ \& \ predicate]$
- The predicate part is optional
- The numeric order of the binding is used to determine the order in the sequence
- The smallest number is taken to be the first index
- Quick examples
 - $[3 * x \mid x \text{ in set } \{0, \dots, 2\}]$
 - $[x \mid x \text{ in set } \{0, \dots, 4\} \ \& \ x > 2]$


Map Operators

dom m	Domain	$(\text{map } A \text{ to } B) \rightarrow \text{set of } A$
rng m	Range	$(\text{map } A \text{ to } B) \rightarrow \text{set of } B$
m1 munion m2	Merge	$(\text{map } A \text{ to } B) * (\text{map } A \text{ to } B) \rightarrow$ $(\text{map } A \text{ to } B)$
m1 ++ m2	Override	$(\text{map } A \text{ to } B) * (\text{map } A \text{ to } B) \rightarrow$ $(\text{map } A \text{ to } B)$
merge ms	Distr. merge	$\text{set of } (\text{map } A \text{ to } B) \rightarrow \text{map } A \text{ to } B$
s <: m	Dom. restr. to	$\text{set of } A * (\text{map } A \text{ to } B) \rightarrow \text{map } A \text{ to } B$
s <-: m	Dom. restr. by	$\text{set of } A * (\text{map } A \text{ to } B) \rightarrow \text{map } A \text{ to } B$
m :> s	Rng. restr. to	$(\text{map } A \text{ to } B) * \text{set of } A \rightarrow \text{map } A \text{ to } B$
m :-> s	Rng. restr. by	$(\text{map } A \text{ to } B) * \text{set of } A \rightarrow \text{map } A \text{ to } B$
m(d)	Map apply	$(\text{map } A \text{ to } B) * A \rightarrow B$
inverse m	Map inverse	$\text{inmap } A \text{ to } B \rightarrow \text{inmap } B \text{ to } A$
m1 = m2	Equality	$(\text{map } A \text{ to } B) * (\text{map } A \text{ to } B) \rightarrow \text{bool}$
m1 <> m2	Inequality	$(\text{map } A \text{ to } B) * (\text{map } A \text{ to } B) \rightarrow \text{bool}$

Mapping Comprehensions

- Using predicates to define mappings implicitly
- In VDM++ formulated like:
 - $\{ \text{maplet} \mid \text{list of bindings \& predicate} \}$
- The predicate part is optional
- Quick examples
 - $\{ i \mid \rightarrow i*i \mid i: \text{nat1} \ \& \ i \leq 4 \}$
 - $\{ i^{**2} \mid \rightarrow i/2 \mid i \text{ in set } \{1, \dots, 5\} \}$

Invariants



Even = **nat**

inv n == n mod 2 = 0

SpecialPair = **nat** * **real** - the first is smallest

inv mk_(n,r) == n < r

DisjointSets = **set of set of** A

inv ss == **forall** s1, s2 **in set** ss &
s1 <> s2 => s1 **inter** s2 = {}

Agenda

- Part 1(9:00 – 10:30) The VDM++ Language
 - ✓ [Introduction](#)
 - ✓ [Access Modifiers and Constructors](#)
 - ✓ [Instance Variables](#)
 - ✓ [Types](#)
 - [Functions](#)
 - [Expressions, Patterns, Bindings](#)
 - [Operations](#)
 - [Statements](#)
 - [Concurrency](#)
- Part 2 (11:00 – 12:30) [VDMTools and VDM++ examples](#)

Function Definitions (1)

- Explicit functions:

$$f: A * B * \dots * Z \rightarrow R1 * R2 * \dots * Rn$$
$$f(a, b, \dots, z) ==$$

expr

pre preexpr(a, b, ..., z)

post postexpr(a, b, ..., z, **RESULT**)

- Implicit functions:

$$f(a:A, b:B, \dots, z:Z) \ r1:R1, \dots, rn:Rn$$

pre preexpr(a, b, ..., z)

post postexpr(a, b, ..., z, r1, ..., rn)

Implicit functions cannot be executed by the VDM interpreter.

Function Definitions (2)

- Extended explicit functions:

```
f(a:A, b:B, ..., z:Z) r1:R1, ..., rn:Rn ==  
  expr  
pre preexpr(a,b,...,z)  
post postexpr(a,b,...,z,r1,...,rn)
```

Extended explicit functions are a non-standard combination of the implicit colon style with an explicit body.

- Preliminary explicit functions:

```
f: A * B * ... * Z -> R1 * R2 * ... * Rn  
f(a,b,...,z) ==  
  is not yet specified  
pre preexpr(a,b,...,z)  
post postexpr(a,b,...,z,RESULT)
```



Quoting pre- and post-conditions

Given an implicit function definition like:

```
ImplFn(n,m: nat, b: bool) r: nat
pre n < m
post if b then n = r else r = m
```

Two extra functions which can be used elsewhere are automatically created:

```
pre_ImplFn: nat * nat * bool -> bool
pre_ImplFn(n,m,b) ==
  n < m;

post_ImplFn: nat * nat * bool * nat -> bool
post_ImplFn(n,m,b,r) ==
  if b
  then n = r
  else r = m
```

Agenda

- Part 1(9:00 – 10:30) The VDM++ Language
 - ✓ [Introduction](#)
 - ✓ [Access Modifiers and Constructors](#)
 - ✓ [Instance Variables](#)
 - ✓ [Types](#)
 - ✓ [Functions](#)
 - [Expressions, Patterns, Bindings](#)
 - [Operations](#)
 - [Statements](#)
 - [Concurrency](#)
- Part 2 (11:00 – 12:30) [VDMTools and VDM++ examples](#)

Expressions

- Let and let-be expressions
- If-then-else expressions
- Cases expressions
- Quantified expressions
- Set expressions
- Sequence expressions
- Map expressions
- Tuple expressions
- Record expressions
- Is expressions
- Define expressions
- Lambda expressions

Special VDM++ Expressions

- New and Self expressions
- Class membership expressions
- Object comparison expressions
- Object reference expressions

Example Let Expressions

- Let expressions are used for naming complex subexpressions:

```
let d = b ** 2 - 4 * a * c
in
    mk_((-b - sqrt(d))/2a, (-b + sqrt(d))/2a)
```
- Let expressions can also be used for breaking down complex data structures into components:

```
let mk_Report(tel, -, ov) = rep
in
    sub-expr
```

Example Let-be expressions

- Let-be-such-that expressions are even more powerful. A free choice can be expressed:

```
let i in set inds l be st Largest(elems l, l(i))
in
  sub_expr
```

and

```
let l in set Permutations(list) be st
  forall i,j in set inds l & i < j => l(i) <= l(j)
in l
```

If-then-else Expressions

If-then-else expressions are similar to those known from programming languages.

```
if c in set dom rq  
then rq(c)  
else {}
```

and

```
if i = 0  
then <Zero>  
elseif 1 <= i and i <= 9  
then <Digit>  
else <Number>
```

Cases Expressions

Cases expressions are very powerful because of pattern matching:

cases com:

```
mk_Loan(a,b) -> a^" has borrowed "^b,  
mk_Receive(a,b) -> a^" has returned "^b,  
mk_Status(l) -> l^" are borrowing "^Borrows(l),  
others -> "some other command is used"
```

end

and

cases a:

```
mk_A(a',-,a') -> Expr(a'),  
mk_A(b,b,c) -> Expr2(b,c)
```

end

Set Expressions

- Set enumeration:
 $\{a, 3, 3, \mathbf{true}\}$
- Set comprehension can either use set binding:
 $\{a+2 \mid \mathbf{mk_}(a, a) \text{ in set } \{\mathbf{mk_}(\mathbf{true}, 1), \mathbf{mk_}(1, 1)\}\}$
or type binding:
 $\{a \mid a: \mathbf{nat} \ \& \ a < 10\}$
- Set range expression:
 $\{3, \dots, 10\}$

Sequence Expressions

- Sequence enumeration:

```
[7.7, true, "I", true]
```

- Sequence comprehension can only use a set bind with numeric values (numeric order is used):

```
[i*i | i in set {1,2,4,6}]
```

and

```
[i | i in set {6,3,2,7} & i mod 2 = 0]
```

- Subsequence expression:

```
[4, true, "string", 9, 4](2, ..., 4)
```

Map Expressions

- Map enumeration:
 $\{1 \mid \rightarrow \mathbf{true}, 7 \mid \rightarrow 6\}$
- Map comprehension can either use type binding:
 $\{i \mid \rightarrow \mathbf{mk_}(i, \mathbf{true}) \mid i: \mathbf{bool}\}$

or set binding:

$$\{a+b \mid \rightarrow b-a \mid a \mathbf{in set} \{1, 2\}, \\ b \mathbf{in set} \{3, 6\}\}$$

and

$$\{i \mid \rightarrow i \mid i \mathbf{in set} \{1, \dots, 10\} \ \& \\ i \mathbf{mod} 3 = 0\}$$

One must be careful to ensure that every domain element maps uniquely to one range element.

Tuple Expressions

- A tuple expression looks like:

`mk_(2 , 7 , true , { | -> })`

- Remember that tuple values from a tuple type will always
 - have the same length and
 - use the same types (possible union types) at corresponding positions.
- On the other hand the length of a sequence value may vary but the elements of the sequence will always be of the same type.

Record Expression

Given two type definitions like:

```
A :: n: nat
    b: bool
    s: set of nat;

B :: n: nat
    r: real
```

one can write expressions like:

```
mk_A(1, true, {8})
mk_B(3, 3)
mu (mk_A(7, false, {1, 4}), n|->1, s|->{})
mu (mk_B(3, 4), r|->5.5)
```

The **mu** operator is called “the record modifier”.

Apply Expressions

- Map applications:

```
let m = { true | -> 5, 6 | -> {} }  
in m(true)
```

- Sequence applications:

```
[ 2, 7, true ] ( 2 )
```

- Field select expressions:

```
let r = mk_A( 2, false, { 6, 9 } )  
in r.b
```

Is Expressions

Basic values and record values can be tested by is- expressions.

`is_nat(5)` will yield true.

`is_C(mk_C(5))` will also yield true, given that C is defined as a record type having one component which 5 belongs to.

`is_A(mk_B(3,7))` will always yield false.

`is_A(6)` will also always yield false.

Define Expressions

The right-hand side of a define expression has access to the instance variables.

The state could be changed by an operation call:

```
def a = OpCall(arg1,arg2) in f(a)
```

or parts of the state could simply be read:

```
def a = instance_variable in g(a)
```


Lambda Expressions

- Lambda expressions are an alternative way of defining explicit functions.

```
lambda n: nat & n * n
```

- They can take a type bind list:

```
lambda a: nat, b: bool &  
  if b then a else 0
```

- or use more complex types:

```
lambda mk_(a, b): nat * nat & a + b
```

New and Self Expressions

- The `new` expression creates an instance of a class and yields a reference to it.
- Given a class called `C` this will create an instance of `C` and return its reference:

```
new C ( )
```

- The `self` expression yields the reference of an object.
- Given a class with instance variable `a` of type `nat` this will initialize an object and yield its reference:

```
Create: nat ==> C  
Create (n) ==  
( a := n;  
  return self )
```

Class Membership Expressions

Check if an object is of a particular class.

```
isofclass(Class_name, object_ref)
```

Returns true if *object_ref* is of class *Class_name* or a subclass of *Class_name*.

Check for the baseclass of a given object.

```
isofbaseclass(Class_name, object_ref)
```

For the result to be true, *object_ref* must be of class *Class_name*, and *Class_name* cannot have any superclasses.

Object Comparison Expressions

Compare two objects.

```
sameclass ( obj1 , obj2 )
```

True if and only if *obj1* and *obj2* are instances of the same class

- sameclass** (m , s) \equiv false
- sameclass** (m , **new** Manager ()) \equiv true

Comparison of baseclasses of two objects.

```
samebaseclass ( obj1 , obj2 )
```

- samebaseclass** (m , s) \equiv true
- samebaseclass** (m , **new** Temporary ()) \equiv false

Object Reference Expressions

- The = and <> operators perform comparison of object references.
- = will only yield true, if the two objects are in fact the same instance.
- <> will yield true, if the two objects are not the same instance, even if they have the same values in all instance variables.

Patterns and Pattern Matching

- Patterns are empty shells
- Patterns are matched thereby binding the pattern identifiers
- There are special patterns for
 - Basic values
 - Pattern identifiers
 - Don't care patterns
 - Sets
 - Sequences
 - Tuples
 - Records

but not for maps

Bindings

- A binding matches a pattern to a value.

- A set binding:

`pat in set expr`

where *expr* must denote a set expression.

pat is bound to the elements of the set *expr*

- A type binding:

`pat : type`

Here *pat* is bound to the elements of *type*.

Type bindings cannot be executed by the interpreter, because such types can be infinitely large.

Agenda

- Part 1(9:00 – 10:30) The VDM++ Language
 - ✓ [Introduction](#)
 - ✓ [Access Modifiers and Constructors](#)
 - ✓ [Instance Variables](#)
 - ✓ [Types](#)
 - ✓ [Functions](#)
 - ✓ [Expressions, Patterns, Bindings](#)
 - [Operations](#)
 - [Statements](#)
 - [Concurrency](#)
- Part 2 (11:00 – 12:30) [VDMTools and VDM++ examples](#)

Operation Definitions (1)

- Explicit operation definitions:

$o: A * B * \dots \implies R$

$o(a, b, \dots) ==$

stmt

pre expr

post expr

- Implicit operations definitions:

$o(a:A, b:B, \dots) r:R$

ext rd ...

wr ...

pre expr

post expr

Operation Definitions (2)

- Preliminary operation definitions:
o: A * B * ... ==> R
o(a,b,...) ==
 is not yet specified
pre expr
post expr
- Delegated operation definitions:
o: A * B * ... ==> R
o(a,b,...) ==
 is subclass responsibility
pre expr
post expr

Operation Definitions (3)

- Operations in VDM++ can be overloaded
- Different definitions of operations with same name
- Argument types must not be overlapping statically (structural equivalence omitting invariants)

Example Operation Definitions

An implicit operation definition could look like:

```
Withdraw(amount: nat) newBalance: int  
ext rd limit    : int  
    wr balance : int  
pre balance - amount > limit  
post balance + amount = balance~ and newBalance = balance
```

An explicit operation definition could look like:

```
Withdraw: nat ==> int  
Withdraw(amount) ==  
( balance := balance - amount;  
  return balance  
)  
pre balance - amount > limit
```

Agenda

- Part 1(9:00 – 10:30) The VDM++ Language
 - ✓ [Introduction](#)
 - ✓ [Access Modifiers and Constructors](#)
 - ✓ [Instance Variables](#)
 - ✓ [Types](#)
 - ✓ [Functions](#)
 - ✓ [Expressions, Patterns, Bindings](#)
 - ✓ [Operations](#)
 - [Statements](#)
 - [Concurrency](#)
- Part 2 (11:00 – 12:30) [VDMTools and VDM++ examples](#)

Statements

- Let and Let-be statements
- Define Statements
- Block statements
- Assign statements
- Conditional statements
- For loop statements
- While loop statements
- Call Statements
- Non deterministic statements
- Return statements
- Exception handling statements
- Error statements
- Identity statements

Agenda

- Part 1(9:00 – 10:30) The VDM++ Language
 - ✓ [Introduction](#)
 - ✓ [Access Modifiers and Constructors](#)
 - ✓ [Instance Variables](#)
 - ✓ [Types](#)
 - ✓ [Functions](#)
 - ✓ [Expressions, Patterns, Bindings](#)
 - ✓ [Operations](#)
 - ✓ [Statements](#)
 - [Concurrency](#)
- Part 2 (11:00 – 12:30) [VDMTools and VDM++ examples](#)



Concurrency Primitives in VDM++ iha.dk

- Concurrency in VDM++ is based on *threads*
- Threads communicate using shared objects
- Synchronization on shared objects is specified using *permission predicates*

Threads

- Modelled by a class with a thread section

```
class SimpleThread
```

```
thread
```

```
    let - = new IO().echo("Hello World!")
```

```
end SimpleThread
```

- Thread execution begins using start statement with an instance of a class with a thread definition


```
start(new SimpleThread)
```

Thread Communication

- Threads operating in isolation have limited use.
- In VDM++ threads communicate using shared objects.

A Producer-Consumer Example

- Concurrent threads must be synchronized
- Illustrate with a producer-consumer example
- Produce before consumption ...
- Assume a single producer and a single consumer
- Producer has a thread which repeatedly places data in a buffer
- Consumer has a thread which repeatedly fetches data from a buffer





The Producer Class

```
class Producer
```

```
instance variables
```

```
b : Buffer
```

```
operations
```

```
Produce: ( ) ==> seq of char
```

```
Produce() == ...
```

```
thread
```

```
  while true do
```

```
    b.Put(Produce())
```

```
end Producer
```

The Consumer Class

```
class Consumer
```

```
instance variables
```

```
b : Buffer
```

```
operations
```

```
Consume: seq of char ==> ( )
```

```
Consume(d) == ...
```

```
thread
```

```
  while true do
```

```
    Consume(b.Get())
```

```
end Consumer
```



The Buffer Class

```
class Buffer

instance variables

data : [seq of char] := nil

operations

public Put: seq of char ==> ()
Put(newData) ==
  data := newData;

public Get: () ==> seq of char
Get() ==
  let oldData = data
  in
  ( data := nil;
    return oldData
  )

end Buffer
```

Permission Predicates

- What if the producer thread generates values faster than the consumer thread can consume them?
- Shared objects require *synchronization*.
- Synchronization is achieved in VDM++ using *permission predicates*.
- A permission predicate describes when an operation call may be executed.
- If a permission predicate is not satisfied, the operation call blocks.

Permission Predicates

- General structure

sync

```
per operation name => predicate;
```

...

- For Put and Get we could write:

```
per Put => data = nil;
```

```
per Get => data <> nil;
```


History Counters and mutex

Counter	Description
#req op	The number of times that op has been requested
#act op	The number of times that op has been activated
#fin op	The number of times that op has been completed
#active op	The number of active executions of op

- Mutual excusion (**mutex**)
- Blocking Puts and Gets while executing:
 - **mutex** (Put , Get)

Permission Predicates: Details

- Permission predicates are described in the sync section of a class

```
sync
```

```
per <operation name> => predicate
```

- The predicate may refer to the class's instance variables.
- The predicate may also refer to special variables known as *history counters*.

History Counters

- History counters provide information about the number of times an operation has been
 - requested
 - activated
 - completed

Counter	Description
#req(op)	The number of times that op has been requested
#act(op)	The number of times that op has been activated
#fin(op)	The number of times that op has been completed
#active(op)	The number of currently active invocations of op (#req - #fin)

The Buffer Synchronized

- Assuming the buffer does not lose data, there are two requirements:
 - It should only be possible to *get* data, when the producer has placed data in the buffer.
 - It should only be possible to *put* data when the consumer has fetched data from the buffer.
- The following permission predicates could model these requirements:
 - `per Put => data = nil`
 - `per Get => data <> nil`

The Buffer Synchronized (2)

- The previous predicates could also have been written using history counters:
- For example
$$\text{per Get} \Rightarrow \#fin(\text{Put}) - \#fin(\text{Get}) = 1$$

Mutual Exclusion


- Another problem could arise with the buffer: what if the producer produces and the consumer consumes at the same time?
- The result could be non-deterministic and/or counter-intuitive.
- VDM++ provides the keyword `mutex`
 - `mutex(Put, Get)`
- Shorthand for
 - `per Put => #active(Get) = 0`
 - `per Get => #active(Put) = 0`

Agenda

- ✓ Part 1(9:00 – 10:30) [The VDM++ Language](#)
- Part 2 (11:00 – 12:30) VDMTools and VDM++ examples
 - [VDMTools overview](#)
 - [The VDM++/UML Process with the alarm example](#)
 - [Industrial usage of VDMTools](#)



VDMTools[®] Overview





iha.dk

Japanese Support via Unicode

The screenshot shows the VDM++ Toolbox E-Lock-Mac.pri application. The main window contains several panels:


- Manager:** A panel with tabs for 'Project' and 'Class'. It has sub-tabs for 'VDM View' and 'Java View'. Below these is a table of classes:

Classes	Syntax	Type	C++	Java	Pretty Print
「錠」	S	T			
「取手」	S	T			
「ボタン」	S	T			
「施錠灯」	S	T			
「表示窓」	S	T			
FSequence	S	T			
IO	S	T			
KeyCommon	S	T			
Store	S	T			

- Source Window:** A window titled 'StoreRoom.vpp' showing the following code:


```
28: public 解錠する : () ==> ()
29: 解錠する() == (
30:   if 表示窓.点いている() and 錠.鍵が一致(表示窓.内容) then
31:     (表示窓.消す());
32:     施錠灯.消す()
33:   else
34:     skip
35:   )
36: pre 表示窓.点いている() and 施錠灯.点いている()
37: post   表示窓.消えている() and 施錠灯.消えている();
38:
39: public 施錠する : () ==> ()
40: 施錠する() == (
41:   if 取手.閉まっている() then
42:     (表示窓.消す());
43:     施錠灯.点ける()
44:   else
45:     skip
46:   )
```

Validation with VDMTools[®]



Documentation in MS Word/RTF

One compound document:



The screenshot shows a Microsoft Word window titled "Microsoft Word - overall.doc". The document content includes:

operations
The InsertCard operation models the activity of inserting a card into the till. This cannot be done if the till holds a card already, which is documented in the precondition.


```
Validate : Card PinCode ==> <PinOk> | <PinNotOk> | <Retained>  
Validate (pin) ==  
  let cardId = curCard.GetCardId(),  
      codeOk = curCard.GetCode() = Encode(pin),  
      cardLegal = IsLegalCard()  
  in  
    (cardOk := codeOk and cardLegal;  
     if not cardLegal then  
       (retainedCards := retainedCards union (curCard),  
        curCard := nil;  
        return <Retained>)  
     elseif codeOk then  
       resource.ResetNumberOfTries (cardId)  
     else  
       (resource.IncrNumberOfTries (cardId);  
        if resource.NumberOfTriesExceeded (cardId) then  
          (retainedCards := retainedCards union (curCard);  
           cardOk := false;  
           curCard := nil;  
           return <Retained>));  
        return if cardOk  
              then <PinOk>  
              else <PinNotOk>)  
  pre CardInside() and not cardOk;
```

The table below presents test coverage information for the Till class.


<i>name</i>	<i>#calls</i>	<i>coverage</i>
Till MakeWithdrawal	1	100%
Till RequestStatement	2	100%
Till ReturnCard	1	100%
Till Validate	9	78%
total		86%

- Documentation
- Specification
- Test coverage
- Test coverage statistics

Architecture of the Rose VDM++ Link



Integrity checker



The screenshot shows the VDM++ Toolbox interface. The main window is titled "The VDM++ Toolbox E-Lock-Mac.prj". It features a toolbar with various icons for file operations and tool execution. Below the toolbar, there are several panels:

- Manager:** A panel with tabs for "Project" and "Class". It shows a list of classes with columns for "Syntax", "Type", "C++", "Java", and "Pretty Print". The classes listed are: 『錠』, 『取手』, 『ボタン』, 『施錠灯』, 『表示窓』, FSequence, IO, KeyCommon, and Store.
- Integrity properties:** A table showing the results of the integrity checker. The table has columns: "Checked", "Module", "Member", "Location", "Index", and "Type".
- Source Window:** A window showing the source code for "StoreRoom.vpp" and "E-Lock.vpp".

Checked	Module	Member	Location	Index	Type
No	『錠』	『錠』	operation	1	state invariants
No	『錠』	『錠』	operation	2	state invariants
No	『錠』	正しい鍵	function	1	subtype
No	『錠』	正しい鍵	function	2	subtype
No	『錠』	正しい鍵	function	3	post condition
No	『錠』	登録する	operation	1	state invariants
No	『取手』	開く	operation	1	subtype

The "Integrity properties" panel also includes a section for "Available" and "Selected" items, with arrows indicating the relationship between them. The "Available" section lists: Available, FSequence, IO, and Store. The "Selected" section lists: 『錠』, 『取手』, 『ボタン』, 『施錠灯』, and 『表示窓』.

```
(forall a鍵 : [ 『錠』 ] &
(let RESULT:bool = a鍵 = nil or
(len a鍵) = 鍵桁数)
in
not (a鍵 = nil) =>
is_(a鍵, seq of nat)))
```

The "Source Window" shows the following code:

```
14: 正しい鍵(a鍵) == a鍵 = nil or (len a鍵 = 鍵桁数)
15: post          a鍵 = nil or (len a鍵 = 鍵桁数);
16:
17: operations
18: public 鍵が一致 : 『錠』 ==> bool
19: 鍵が一致(a鍵) == return 施錠鍵 = a鍵;
```

Agenda

- ✓ Part 1(9:00 – 10:30) [The VDM++ Language](#)
- Part 2 (11:00 – 12:30) VDMTools and VDM++ examples
 - ✓ [VDMTools overview](#)
 - [The VDM++/UML Process with the alarm example](#)
 - [Industrial usage of VDMTools](#)




Steps to Develop a Formal Model iha.dk

1. Determine the purpose of the model.
2. Read the requirements.
3. Analyze the functional behavior from the requirements.
4. Extract a list of possible classes or data types (often from nouns) and operations (often from actions). Create a dictionary by giving explanations to items in the list.
5. Sketch out representations for the classes using UML class diagrams. This includes the attributes and the associations between classes. Transfer this model to VDM++ and check its internal consistency.
6. Sketch out signatures for the operations. Again, check the model's consistency in VDM++.
7. Complete the class (and data type) definitions by determining potential invariant properties from the requirements and formalizing them.
8. Complete the operation definitions by determining pre- and post conditions and operation bodies, modifying the type definitions if necessary.
9. Validate the specification using systematic testing and rapid prototyping.
10. Implement the model using automatic code generation or manual coding.

A Chemical Plant



iha.dk





A Chemical Plant Requirements

1. A computer-based system is to be developed to manage the alarms of this plant.
2. Four kinds of qualifications are needed to cope with the alarms: electrical, mechanical, biological, and chemical.
3. There must be experts on duty during all periods allocated in the system.
4. Each expert can have a list of qualifications.
5. Each alarm reported to the system has a qualification associated with it along with a description of the alarm that can be understood by the expert.
6. Whenever an alarm is received by the system an expert with the right qualification should be found so that he or she can be paged.
7. The experts should be able to use the system database to check when they will be on duty.
8. It must be possible to assess the number of experts on duty.



The Purpose of the VDM++ Model iha.dk

The **purpose** of the model is to clarify the rules governing the duty roster and calling out of experts to deal with alarms.



Creating a Dictionary

- Potential Classes and Types (Nouns)
 - **Alarm**: required qualification and description
 - **Plant**: the entire system
 - **Qualification** (electrical, mechanical, biological, chemical)
 - **Expert**: list of qualifications
 - **Period** (whatever shift system is used here)
 - **System and system database**? This is probably a kind of schedule.
- Potential Operations (Actions)
 - **Expert to page**: when an alarm appears (what's involved? Alarm operator and system)
 - **Expert is on duty**: check when on duty (what's involved? Expert and system)
 - **Number of experts on duty**: presumably given period (what's involved? operator and system)

Guideline 1

Nouns from a dictionary should be modeled as types if, for the purposes of the model, they need have only trivial functionality in addition to read/write.

Alarm

 reqQuali : Expert`Qualification
 descr : String


Expert

 quali : Qualification

Sketching an Alarm

Defined as a VDM++ class:

```
class Alarm
instance variables
  reqQuali: Expert`Qualification
  descr    : String;
end Alarm
```



Alternative Alarm

Alarm could also have been defined as a composite type:

```
Alarm :: reqQuali : Expert`Qualification  
       descr      : String
```

Then if *a* is of type *Alarm*:


a.descr is the description of *a*

```
a.descr : String
```

```
a.reqQuali : Expert`Qualification
```

Guideline 2

Create an overall class to represent the entire system so that the precise relationships between the different classes and their associations can be expressed there.




Guideline 3 and 4

Whenever an association is introduced consider its multiplicity and give it a rôle name in the direction in which the association is to be used.

If an association depends on some value, a qualifier should be introduced for the association. The name of the qualifier must be a VDM++ type.

Initial Class Diagram



```
class Plant
instance variables
public alarms : set of Alarm;
public schedule : map Period to set of Expert;

end Plant
```

Guideline 5

Declare instance variables to be **private** or **protected** to keep encapsulation. If nothing is specified by the user, **private** is assumed automatically.

```
class Expert
instance variables
private quali: set of Qualification;
end Expert
```

```
class Alarm
instance variables
private descr    : String;
private reqQuali: Qualification;
end Alarm
```

Guideline 6 and 7

Use VDMTools to check internal consistency as soon as class skeletons have been completed and before any functionality has been introduced.

- Definition of types missing
- To be updated in the respective classes
- Resynchronized with the UML model

```
class Plant
types
  Period = token;
end Plant
```

Tokens are useful for abstract models where unspecified values are to be used.




Adding Quantification and String

```
class Expert
types
  Qualification = <Mech> | <Chem> | <Bio> | <Elec>
end Expert
```


```
class Alarm
types
public String = seq of char;
instance variables
  descr      : String;
  reqQuali   : Expert`Qualification;
end Alarm
```

Guideline 8

Think carefully about the parameter types and the result type as this often helps to identify missing connections in the class diagram.



Updated UML Class Diagram



Guideline 9

Document important properties or constraints as invariants.

```
class Plant
```

```
...
```

```
instance variables
```

```
alarms : set of Alarm;
```

```
schedule: map Period to set of Expert;
```

```
inv forall p in set dom schedule & schedule(p) <> {};
```

```
end Plant
```

Guideline 10

When there are several alternative ways of performing some functionality, use an implicit definition so that subsequent development work is not biased.

```
ExpertToPage: Alarm * Period ==> Expert
ExpertToPage(a, p) ==
  is not yet specified
pre a in set alarms and
  p in set dom schedule
post let expert = RESULT
  in
  expert in set schedule(p) and
  a.GetReqQuali() in set expert.GetQuali();
```


Will the Qualification exist?

- How can we be sure that an expert with the required qualification exists in the required period?
- We need to add an invariant to the instance variables of the *Plant* class
- That is using guideline 11

Guideline 11

When defining operations, try to identify additional invariants.

instance variables

```
alarms : set of Alarm;  
schedule: map Period to set of Expert;  
inv forall p in set dom schedule & schedule(p) <> {};  
inv forall a in set alarms &  
    forall p in set dom schedule &  
        exists expert in set schedule(p) &  
            a.GetReqQuali() in set expert.GetQuali();
```



Further Operations inside Plant

```
class Plant
operations
...

public NumberOfExperts: Period ==> nat
NumberOfExperts(p) ==
    return card schedule(p)
pre p in set dom schedule;

public ExpertIsOnDuty: Expert ==> set of Period
ExpertIsOnDuty(ex) ==
    return {p | p in set dom schedule &
             ex in set schedule(p)};

end Plant
```

Guideline 12

Try to make explicit operation definitions precise and clear and yet abstract compared to code written in a programming language.

```
import java.util.*;

class Plant {


    Map schedule;

    Set ExpertIsOnDuty(Integer ex) {
        TreeSet resset = new TreeSet();
        Set keys = schedule.keySet();
        Iterator iterator = keys.iterator();

        while(iterator.hasNext()) {
            Object p = iterator.next();
            if ( (Set) schedule.get(p).contains(ex))
                resset.add(p);
        }
        return resset;
    }
}
```



Final UML Class Diagram



Guideline 13

Whenever a class has an invariant on its instance variables and it has a constructor, it is worth placing the invariant in a separate function if the constructor needs to assign values to the instance variables involved in the invariant.

functions

```
PlantInv: set of Alarm * map Period to set of Expert ->  
         bool  
PlantInv(as, sch) ==  
  (forall p in set dom sch & sch(p) <> {}) and  
  (forall a in set as &  
   forall p in set dom sch &  
    exists expert in set sch(p) &  
    a.GetReqQuali() in set expert.GetQuali());
```

To be used inside Plant Constructor

```
class Plant
...
public Plant: set of Alarm *
           map Period to set of Expert ==>
           Plant
Plant(als,sch) ==
( alarms := als;
  schedule := sch
)
pre PlantInv(als,sch);
end Plant
```

Review Requirements [1]

R1: A computer-based system managing this plant is to be developed.

Considered in the `Plant` class definition and the operation and function definitions.

R2: Four kinds of qualifications are needed to cope with the alarms: electrical, mechanical, biological, and chemical.

Considered in the `Qualification` type definition of the `Expert` class.

R3: There must be experts on duty at all times during all periods which have been allocated in the system.

Invariant on the instance variables of class `Plant`.

Review Requirements (2)

R4: Each expert can have a list of qualifications.

Assumption: non-empty set instead of list in class Expert.

R5: Each alarm reported to the system must have a qualification associated with it and a description which can be understood by the expert.

Considered in the instance variables of the Alarm class definition assuming that it is precisely one qualification.

R6: Whenever an alarm is received by the system an expert with the right qualification should be paged.

The ExpertToPage operation with additional invariant on the instance variables of the Plant class definition.

Review the Requirements (3)

R7: The experts should be able to use the system database to check when they will be on duty.

The ExpertOnDuty operation.

R8: It must be possible to assess the number of experts on duty.

The NumberOfExperts with assumption for a given period.

Agenda

- ✓ Part 1(9:00 – 10:30) [The VDM++ Language](#)
- Part 2 (11:00 – 12:30) VDMTools and VDM++ examples
 - ✓ [VDMTools overview](#)
 - ✓ [The VDM++/UML Process with the alarm example](#)
 - [Industrial usage of VDMTools](#)

ConForm (1994)

- Organisation: British Aerospace (UK)
- Domain: Security (gateway)
- Tools: The IFAD VDM-SL Toolbox
- Experience:
 - Prevented propagation of error
 - Successful technology transfer
 - At least 4 more applications without support
- Statements:
 - “Engineers can learn the technique in one week”
 - “**VDMTools**[®] can be integrated gradually into a traditional existing development process”

DustExpert (1995-7)

- Organisation: Adelard (UK)
- Domain: Safety (dust explosives)
- Tools: The IFAD VDM-SL Toolbox
- Experience:
 - Delivered on time at expected cost
 - Large VDM-SL specification
 - Testing support valuable
- Statement:
 - “Using **VDMTools**[®] we have achieved a productivity and fault density far better than industry norms for safety related systems”



Adelard Metrics

Initial requirements	450 pages
VDM specification	16kloc (31 modules) 12kloc (excl comments)
Prolog implementation	37kloc 16kloc (excl comments)
C++ GUI implementation	23kloc 18kloc (excl comments)

- 31 faults in Prolog and C++ ($< 1/\text{kloc}$)
- Most minor, only 1 safety-related
- 1 (small) design error, rest in coding

CAVA (1998-2000)

- Organisation: Baan (Denmark)
- Domain: Constraint solver (Sales Configuration)
- Tools: The IFAD VDM-SL Toolbox
- Experience:
 - Common understanding
 - Faster route to prototype
 - Earlier testing
- Statement:
 - “**VDMTools**[®] has been used in order to increase quality and reduce development risks on high complexity products”

Dutch DoD (1997-8)

- Organisation: Origin, The Netherlands
- Domain: Military
- Tools: The IFAD VDM-SL Toolbox
- Experience:
 - Higher level of assurance
 - Mastering of complexity
 - Delivered at ***expected cost*** and ***on schedule***
 - ***No errors detected in code after delivery***
- Statement:
 - “We chose **VDMTools**[®] because of high demands on maintainability, adaptability and reliability”

DoD, NL Metrics (1)


	kloc	hours	loc/hour
spec	15	1196	13
manual impl	4	471	8.5
automatic impl	90	0	NA
test	NA	612	NA
total code	94	2279	41.2

- Estimated 12 C++ loc/h with manual coding!




DoD - Comparative Metrics

Traditional:



VDMTools[®]:



BPS 1000 (1997-)

- Organisation: GAO, Germany
- Domain: Bank note processing
- Tools: The IFAD VDM-SL Toolbox
- Experience:
 - Better understanding of sensor data
 - Errors identified in other code
 - Savings on maintenance
- Statement:
 - VDMTools provides unparalleled support for design abstraction ensuring quality and control throughout the development life cycle.

Flower Auction (1998)

- Organisation: Chess, The Netherlands
- Domain: Financial transactions
- Tools: The IFAD VDM++ Toolbox
- Experience:
 - Successful combination of UML and VDM++
 - Use iterative process to gain client commitment
 - Implementers did not even have a VDM course
- Statement:
 - “The link between VDMTools and Rational Rose is essential for understanding the UML diagrams”

SPOT 4 (1999)

- Organisation: CS-CI, France
- Domain: Space (payload for SPOT4 satellite)
- Tools: The IFAD VDM-SL Toolbox
- Experience:
 - 38 % less lines of source code
 - 36 % less overall effort
 - Use of automatic C++ code generation
- Statement:

The cost of applying Formal methods is significantly lower than without them.

Japanese Railways (2000-2001)



- Domain: Railways (database and interlocking)
- Experience:
 - Prototyping important
 - Now also using it for ATC system
- Engineer working at IFAD for two years with PROSPER proof support

Stock-options (2000-)

- Organisation: JFITS (CSK group company), Japan
- Domain: Financial
- Tools: The IFAD VDM++ Toolbox
- Reason for CSK to purchase VDMTools

Tax exemption	COCOMO	Realized
Effort	38,5 person months	14 person months
Schedule	9 months	3,5 months

Options	COCOMO	Realized
Effort	147,2 person months	60,1 person months
Schedule	14,3 months	7 months

Reverse Engineering (2001)

- Organisation: Boeing
- Domain: Avionics
- Tools: The IFAD VDM++ Toolbox
- Included development of Java to VDM++ reverse engineering feature


Optimisation (2001)

- Organisation: Transitive Technologies, UK
- Domain: Embedded
- Tools: The IFAD VDM-SL Toolbox
- Making software independent of hardware platform

Quote of the day

The successful construction of all machinery depends on the perfection of the tools employed, and whoever is the master in the art of tool-making possesses the key to the construction of all machines.

Charles Babbage, 1851



Any questions?